รายการอ้างอิง

1. พีรพัฒน์ ศิริสมบูรณ์ลาภ. กลศาสตร์ท้องฟ้าเบื้องต้น. พิมพ์ครั้งที่ 1. กรุงเทพมหานคร : สำนักพิมพ์จุฬาลงกรณ์มหาวิทยาลัย, 2534.
2. พีรพัฒน์ ศิริสมบูรณ์ลาภ. กลศาสตร์ท้องฟ้าทั่วไป. พิมพ์ครั้งที่ 1. กรุงเทพมหานคร : สำนักพิมพ์จุฬาลงกรณ์มหาวิทยาลัย, 2533.
3. Danby, J.M.A. Fundamental of Celestial Mechanics. 2nd Edition. Virginia : Willmann-Bell, 1992.
4. Boulet, D.L. Method of Orbit Determination for the Microcomputer. Virginia : Willmann-Bell Inc., 1991.
5. Roy, A.E. Orbital Motion. Bristol : Adam Hilger Ltd., 1980.
6. Smart, W.M. Text Book on Spherical Astronomy. Cambridge : Cambridge University Press, 1971.
7. Dancan, J.C. Astronomy. New York and London : Harper & Brothers Publish, 1950.
8. Rika Nempyoo Heisei Kyu Nen. Chronological Scientific.Table 2000. Tokyo : Maruzen National Astronomical Observatory, 2001.
9. Woolard, E.W. and Clemence,G.M. Spherical Astronomy. New York : Acadmic Press, 1966.
10. Mounton, F.R. An Introduction to Celestial Mechanics. New York : Dover, 1970.
11. Murray, C.D. and Dermott. Solar System Dynamics.1st Edition Cambridge : Cambridge University Press, 1999.
12. Boccaletti, D and Pucacco, G. Theory of Orbits. New York : Springer-Verlag, 1996.
13. Isaacs, A. Dictionary of Physics.3rd Edition New York : Mackays PLC, 1996.
14. Serway, R.A. Physics for Science & Engineering with Modern Physics. 4thEdition New York : Saunders College, 1996.
15. Fowles, G.R. and Cassiday, G.L. Analytical Mechanics. 5thEdition New York : Harcourt Brace College, 1996.
16. Escobal, P.R. Method of Orbit Determination. New York : Krieger Publishing, 1976.
17. Illingworth,V. The Facts on File Dictionary of Astronomy.3sd Edition U.S.A : Fact on File Inc, 1994.

ภาคผนวก

ภาคผนวก ก

ค่าคงที่ทางดาราศาสตร์

1 หน่วยดาราศาสตร์ (AU) =1.4959787×10^{11} เมตร

1 หน่วยมวลดวงอาทิตย์ = 1.9891×10^{30} กิโลกรัม

1 วัน = 86,400 วินาที

1 ปี = 365.2425 วัน

ค่าคงที่แรงโน้มถ่วงสากลของเกาส์ (k) =0.01720209895 $AU^3.Solar\ mas^{-1}.day^{-2}$

มวลของโลก = 0.000003003 หน่วยมวลของดวงอาทิตย์

มวลของดาวพฤหัสบดี = 0.000954791 หน่วยมวลของดวงอาทิตย์

มวลของดาวเสาร์ = 0.000285878 หน่วยมวลของดวงอาทิตย์

ภาคผนวก ข

โปรแกรมการคำนวณตำแหน่งและความเร็วที่เวลาเริ่มต้น

```
1000 CLS
1010 PRINT "# POSVEL # POSITION AND VELOCITY"
1020 PRINT "# FROM CLASSICAL ELEMENTS"
1030 REM
1040 DEFDBL A-Z
1050 DEFINT K
1060 REM
1070 DEF FNSH (X) = (EXP(X) - EXP(-X)) / 2
1080 DEF FNCH (X) = (EXP(X) + EXP(-X)) / 2
1090 REM
1100 REM
1110 Q1 = .0174532925#
1120 G$ = "#######.#######"
1130 S$ = "#########.#####"
1140 AQ$ = "A"
1150 MT$ = "MM"
1160 REM
1170 READ EC
1180 READ N$, E$, K#
1190 READ M
1200 READ T0
1210 READ AQ, E, MT
1220 READ I, OO, W
1230 REM
1240 IF AQ < 0 THEN MT$ = "TT"
1250 IF ABS(1 - E) < .001# THEN AQ$ = "Q"
1260 IF AQ$ = "Q" THEN MT$ = "TT"
1270 REM
```

```
1280 LINE INPUT ""; A$
1290 CLS
1300 PRINT "POSITION AND VELOCITY"
1310 PRINT "FROM"
1320 PRINT "CLASSICAL ELEMENT"
1330 PRINT N$
1340 PRINT E$
1350 PRINT "K"; TAB(7); : PRINT USING G$; K#
1360 PRINT "EC"; TAB(7); : PRINT USING G$; EC
1370 PRINT
1380 PRINT "T(0)"; TAB(7); : PRINT USING S$; T0
1390 PRINT
1400 PRINT "CLASSICAL ELEMENT"
1410 PRINT
1420 PRINT AQ$; TAB(7); : PRINT USING G$; AQ
1430 PRINT "E"; TAB(7); : PRINT USING G$; E
1440 PRINT MT$; TAB(7); : PRINT USING S$; MT
1450 PRINT "I"; TAB(7); : PRINT USING S$; I
1460 PRINT "OO"; TAB(7); : PRINT USING S$; OO
1470 PRINT "W"; TAB(7); : PRINT USING S$; W;
1480 REM
1490 LINE INPUT ""; A$
1500 CLS
1510 EC = EC * Q1
1520 I = I * Q1
1530 OO = OO * Q1
1540 W = W * Q1
1550 REM
1560 REM
1640 REM *** ELLIPTIC MOTION ***
1650 REM
1660 MM = MT * Q1
```

```
1670 EE = MM
1680   F = EE - E * SIN(EE) - MM
1690   PRINT "F=";
1700   PRINT USING G$; F
1710   IF ABS(F) < .0000001# THEN 1750
1720   DF = 1 - E * COS(EE)
1730   EE = EE - F / DF
1740   GOTO 1680
1750 R = AQ * (1 - E * COS(EE))
1760 EP = SQR(M / AQ) / R
1770 B = AQ * SQR(1 - E * E)
1780 XB = AQ * (COS(EE) - E)
1790 YB = B * SIN(EE)
1800 XP = -AQ * EP * SIN(EE)
1810 YP = B * EP * COS(EE)
1820 GOTO 2260
1830 REM
2240 REM ***UNIT VECTORS PP AND QQ ***
2250 REM
2260 PP(1) = COS(W) * COS(OO) - SIN(W) * SIN(OO) * COS(I)
2270 PP(2) = COS(W) * SIN(OO) + SIN(W) * COS(OO) * COS(I)
2280 PP(3) = SIN(W) * SIN(I)
2290 REM
2300 QQ(1) = -SIN(W) * COS(OO) - COS(W) * SIN(OO) * COS(I)
2310 QQ(2) = -SIN(W) * SIN(OO) + COS(W) * COS(OO) * COS(I)
2320 QQ(3) = COS(W) * SIN(I)
2330 REM
2340 REM *** POSITION AND VELOCITY ***
2350 REM
2360 FOR K = 1 TO 3
2370   R(K) = XB * PP(K) + YB * QQ(K)
2380   V(K) = XP * PP(K) + YP * QQ(K)
```

```
2390 NEXT K
2400 REM
2410 REM *** REDUCTION TO EQUATION IF REQUIRED ***
2420 REM
2430 R0(1) = R(1)
2440 R0(2) = R(2) * COS(EC) - R(3) * SIN(EC)
2450 R0(3) = R(3) * COS(EC) + R(2) * SIN(EC)
2460 REM
2470 V0(1) = V(1)
2480 V0(2) = V(2) * COS(EC) - V(3) * SIN(EC)
2490 V0(3) = V(3) * COS(EC) + V(2) * SIN(EC)
2500 REM
2510 LINE INPUT ""; A$
2520 CLS
2530 PRINT "POSITION AND VELOCITY"
2540 PRINT
2550 PRINT "R(K)"; TAB(7);
2560 PRINT USING G$; R0(1); R0(2); R0(3)
2570 PRINT "V(K)"; TAB(7);
2580 PRINT USING G$; V0(1); V0(2); V0(3)
2590 END
20000 'DATA  23.439291#
20010 'DATA "THULLE","J2000.0",0.017202099#
20020 'DATA 1
20030 'DATA 2452200.5#
20040 'DATA 4.283#,0.011#,146.6#
20050 'DATA 2.3#,73.7#,77.5#
20060 'REM
30000 'DATA  23.439291#
30010 'DATA"JUPITER","J2000.0",0.017202099#
30020 'DATA 1.000954791#
30030 'DATA 2452200.5#
```

```
30040 'DATA 5.2026#,0.0485#,65.426#
30050 'DATA 1.303#,100.469#,273.866#
30060 'REM
40000 DATA  23.439291#
40010 DATA "SATURN","J2000.0",0.017202099#
40020 DATA 1.000285878#
40030 DATA 2452200.5#
40040 DATA 9.5549#,0.0555#,335.298#
40050 DATA 2.489#,113.661#,67.730285#
40060 REM
50000 DATA  23.439291#
50010 DATA "HIDALGO","J2000.0",0.017202099#
50020 DATA 1
50030 DATA 2452200.5#
50040 DATA 5.750#,0.661#,274.9#
50050 DATA 42.6#,21.6#,56.5#
```

ภาคผนวก ค

โปรแกรมการหาตำแหน่งและความเร็วของดาวเคราะห์น้อยที่เวลาต่างๆโดยระเบียบวิธีของเองเก

```
1000 CLS
1010 PRINT "ENCKE # RA AND DEC"
1020 PRINT "# FROM PERTURED ORBIT MOTION"
1030 PRINT "# BY UNIVERSAL VARIABLE"
1040 PRINT "# AND RUNKE-KUTTA FIVE"
1050 PRINT
1060 DEFDBL A-Z
1070 DEFINT I-K, N, Q
1080 REM
1090 DEF FNVS (X, Y, Z) = X * X + Y * Y + Z * Z
1100 DEF FNMG (X, Y, Z) = SQR(X * X + Y * Y + Z * Z)
1110 DEF FNDP (X1, Y1, Z1, X2, Y2, Z2) = X1 * X2 + Y1 * Y2 + Z1 * Z2
1120 DEF FNASN (X) = ATN(X / SQR(-X * X + 1))
1130 DEF FNACN (X) = -ATN(X / SQR(-X * X + 1)) + 1.507963263#
1140 REM
1150 DEF FNAK (M, PK, P, RK, R) = M * (PK / (P * P * P) - RK / (R * R * R))
1160 DEF FNV (VK) = VK
1170 DEF FNA (M, RK, R, AK) = -(1 + M) * RK / (R * R * R) + AK
1180 DEF FNF (VK) = VK
1190 DEF FNG (M, RU, R1, RP, R2, AK) = (1 + M) * (RU / (R1 * R1 * R1) - RP / (R2 * R2 *
R2)) + AK
1200 REM
1210 DIM T(1000), B(19)
1220 DIM R(1000, 3), V(1000, 3)
1230 REM
1240 Q1# = .017453293#
1250 G$ = "####.#######"
1260 j$ = "####.#####"
```

```
1270 A$ = "#####.######"
1280 D$ = "######.#####"
1290 AM$ = "###"
1300 AC$ = "####.##"
1310 DM$ = "######"
1320 DC$ = "####.#"
1330 REM
1340 READ AB
1350 REM
1360 READ N$(0), E$, K#
1370 READ M(0)
1380 READ T(0)
1390 READ R(0, 1), R(0, 2), R(0, 3)
1400 READ V(0, 1), V(0, 2), V(0, 3)
1410 REM
1420 READ TF
1430 REM
1440 READ RR(1), RR(2), RR(3)
1450 REM
1460 REM
1470 REM
1480 READ NQ, C$, TQ
1490 REM
1500 FOR Q = 1 TO NQ
1510    READ N$(Q)
1520    READ M(Q)
1530    READ RE(Q, 1), RE(Q, 2), RE(Q, 3)
1540    READ VE(Q, 1), VE(Q, 2), VE(Q, 3)
1550 NEXT Q
1560 REM
1570 REM
1580 REM
```

```
1590 PRINT N$(0); C$
1600 PRINT "PERTURBER EPOCH:"; TAB(19); TQ
1610 PRINT "INITIAL EPOCH:"; TAB(20);
1620 PRINT USING j$; T(0)
1630 PRINT "FINAL EPOCH:"; TAB(20);
1640 PRINT USING j$; TF
1650 PRINT "ELAPSED TIME:"; TAB(20);
1660 PRINT USING j$; (TF - T(0))
1670 PRINT
1680 INPUT ">>>NUMBER OF STEPS"; NS
1690 REM
1700 REM
1710 REM
1720 B(1) = 1
1730 FOR j = 2 TO 19
1740   B(j) = B(j - 1) / j
1750 NEXT j
1760 REM
1770 REM
1780 REM
1790 FOR Q = 1 TO NQ
1800   REM
1810   REM * PERTURBER ELEMENTS *
1820   REM
1830   FOR K = 1 TO 3
1840       VE(Q, K) = VE(Q, K) / K#
1850   NEXT K
1860   REM
1870   M = 1 + M(Q)
1880   RV = FNDP(RE(Q, 1), RE(Q, 2), RE(Q, 3), VE(Q, 1), VE(Q, 2), VE(Q, 3))
1890   V2 = FNVS(VE(Q, 1), VE(Q, 2), VE(Q, 3))
1900   R0(Q) = FNMG(RE(Q, 1), RE(Q, 2), RE(Q, 3))
```

```
1910   D0(Q) = RV / SQR(M)
1920   AI(Q) = 2 / R0(Q) - V2 / M
1930   C0(Q) = 1 - R0(Q) * AI(Q)
1940   S0(Q) = D0(Q) * SQR(ABS(AI(Q)))
1950   N#(Q) = K# * SQR(M) * SQR(ABS(AI(Q) * AI(Q) * AI(Q)))
1960   REM
1970 NEXT Q
1980 REM
1990 REM
2000 CLS
2010 PRINT "RA AND DEC"
2020 PRINT "FROM PERTURBED ORBITAL MOTION"
2030 PRINT "BY UNIVERSAL VARIABLE"
2040 PRINT "AND RUNGE-KUTTA FIVE"
2050 PRINT N$(0)
2060 PRINT E$
2070 PRINT "NQ"; NQ; C$; TQ
2080 REM
2090 PRINT TAB(4); "T(I)";
2100 PRINT TAB(18); "X"; TAB(30); "Y"; TAB(42); "Z";
2110 PRINT TAB(54); "R"
2120 REM
2130 REM
2140 REM
2150 TN = TF
2160 DT = (TN - T(0)) / NS
2170 REM
2172 REM
2174 INPUT "enter filename :"; N$
2176 OPEN N$ FOR OUTPUT AS #1
2178 PRINT #1,
2180 FOR i = 0 TO NS
```

```
2190  REM
2200  IF i = 0 THEN 2650
2210  REM
2220  T(i) = T(i - 1) + DT
2230  REM
2240  REM
2250  REM
2260      REM
2270      REM * REFERENCE ORBIT *
2280      REM
2290      GOSUB 11010: REM RORB/SUB
2300      REM
2310      FOR K = 1 TO 3
2320          R(i, K) = F * R(i - 1, K) + G * V(i - 1, K)
2330          V(i, K) = FP * R(i - 1, K) + GP * V(i - 1, K)
2340          RQ(0, K) = R(i - 1, K)
2350          VQ(0, K) = V(i - 1, K)
2360      NEXT K
2370  REM
2380  REM
2390  REM
2400  FOR Q = 1 TO NQ
2410      REM
2420      REM * PERTURBER ORBITS *
2430      REM
2440      GOSUB 12010: REM PORB/SUB
2450      REM
2460      FOR K = 1 TO 3
2470          RQ(Q, K) = F * RE(Q, K) + G * VE(Q, K)
2480          VQ(Q, K) = FP * RE(Q, K) + GP * VE(Q, K)
2490      NEXT K
2500      REM
```

```
2510   NEXT Q
2520   REM
2530   REM
2540   REM
2550        REM
2560        REM * NUMERICAL INTEGRATION *
2570        REM
2580        GOSUB 13010: REM ENK5/SUB
2590        REM
2600        FOR K = 1 TO 3
2610             R(i, K) = R(i, K) + DR(0, K)
2620             V(i, K) = V(i, K) + DV(0, K)
2630        NEXT K
2640        REM
2650        R = FNMG(R(i, 1), R(i, 2), R(i, 3))
2660        REM
2670   PRINT USING j$; T(i);
2680   PRINT USING G$; R(i, 1); R(i, 2); R(i, 3); R;
2682   PRINT TAB(11);
2684   PRINT USING G$; V(i, 1); V(i, 2); V(i, 3)
2690   REM
2700   'LINE INPUT
2702 PRINT #1, USING j$; T(i);
2704 PRINT #1, USING G$; R(i, 1); R(i, 2); R(i, 3); R;
2706 PRINT #1, USING G$; V(i, 1); V(i, 2); V(i, 3)
2708 'LINE INPUT " "; l$
2710 NEXT i
2715 CLOSE
2720 REM
2730 i = i - 1
2740 REM
2750 PRINT TAB(11);
```

```
2760 PRINT USING G$; V(i, 1); V(i, 2); V(i, 3)
2770 REM
2780 REM
2790 REM
2800   REM
2810   REM * LIGHT-TIME *
2820   REM
2830   FOR K = 1 TO 3
2840         P(K) = R(i, K) + RR(K)
2850   NEXT K
2860   REM
2870   P = FNMG(P(1), P(2), P(3))
2880   REM
2890   AP = AB * P
2900   IF T(i) - (TF - AP) < .00001 THEN 3070
2910   T(i) = TF - AP
2920   DT = T(i) - T(i - 1)
2930   REM
2940   GOSUB 11010: REM RORB/SUB
2950   REM
2960   FOR K = 1 TO 3
2970         R(i, K) = F * R(i - 1, K) + G * V(i - 1, K)
2980   NEXT K
2990   REM
3000   GOTO 2830
3010   REM
3020   REM
3030   REM
3040         REM
3050         REM * RA AND DEC *
3060         REM
3070         FOR K = 1 TO 3
```

```
3080            LL(K) = P(K) / P
3090        NEXT K
3100        CD = SQR(1 - LL(3) * LL(3))
3110        CX = LL(1) / CD
3120        SX = LL(2) / CD
3130        REM
3140        GOSUB 16010: REM ARC/SUM
3150        REM
3160        A = X / (15 * Q1#)
3170        D = FNASN(LL(3)) / Q1#
3180        REM
3190        AH = FIX(A)
3200        AM = FIX((A - AH) * 60)
3210        AC = (A - AH - AM / 60) * 3600
3220        REM
3230        DD = FIX(D)
3240        DM = FIX((D - DD) * 60)
3250        DC = (D - DD - DM / 60) * 3600
3260        REM
3270        REM
3280        REM
3290    PRINT
3300    PRINT TAB(4); "T(I)";
3310    PRINT TAB(18); "A(I)"; TAB(30); "D(I)"; TAB(42); "P(I)"
3320    REM
3330    PRINT USING j$; TF;
3340    PRINT USING A$; A;
3350    PRINT USING D$; D;
3360    PRINT USING G$; P
3370    REM
3380    PRINT TAB(13);
3390    PRINT USING AM$; AM;
```

```
3400   PRINT USING AC$; AC;
3410   PRINT USING DM$; DM;
3420   PRINT USING DC$; DC
3430   PRINT
3440   PRINT "ENCKE"
3450   END
11000 STOP
11010 REM # RORB/SUB # COMPUTES REFERENCE ORBIT
11020 REM # UNIVERSAL VARIABLE
11030 M = 1 + M(0)
11040 R0 = FNMG(R(i - 1, 1), R(i - 1, 2), R(i - 1, 3))
11050 D0 = FNDP(R(i - 1, 1), R(i - 1, 2), R(i - 1, 3), V(i - 1, 1), V(i - 1, 2), V(i - 1, 3)) / SQR(M)
11060 AI = 2 / R0 - FNVS(V(i - 1, 1), V(i - 1, 2), V(i - 1, 3)) / M
11070 C0 = 1 - R0 * AI
11080 WW = K# * SQR(M) * DT
11090 XX = WW / R0
11100  X2 = XX * XX
11110  XA = X2 * AI
11120  X3 = X2 * XX
11130  CC = -XA * (B(10) - XA * (B(12) - XA * (B(14) - XA * (B(16) - XA * (B(18))))))
11135  CC = X2 * (B(2) - XA * (B(4) - XA * (B(6) - XA * (B(8) + CC))))
11140  UU = -XA * (B(11) - XA * (B(13) - XA * (B(15) - XA * (B(17) - XA * (B(19))))))
11145  UU = X3 * (B(3) - XA * (B(5) - XA * (B(7) - XA * (B(9) + UU))))
11150  SS = XX - UU * AI
11160  FX = R0 * XX + C0 * UU + D0 * CC - WW
11170  IF ABS(FX) < .00000005# THEN 11210
11180  DF = R0 + CO * CC + D0 * SS
11190  XX = XX - FX / DF
11200  GOTO 11100
11210 F = 1 - CC / R0
11220 G = (R0 * SS + D0 * CC) / SQR(M)
11230 R = R0 + C0 * CC + D0 * SS
```

```
11240 FP = -SQR(M) * SS / (R * R0)
11250 GP = 1 - CC / R
11260 RETURN
12000 STOP
12010 REM # PORB/SUB # COMPUTES PERTURBER ORBIT
12020 REM # CLOSED ELLIPTIC F&G EXPRESSIONS
12030 WQ = N#(Q) * (T(i - 1) - TQ)
12040 GG = WQ
12050   FG = GG - C0(Q) * SIN(GG) - S0(Q) * COS(GG) + S0(Q) - WQ
12060   IF ABS(FG) < .000002# THEN 12100
12070   DF = 1 - C0(Q) * COS(GG) + SO(Q) * SIN(GG)
12080   GG = GG - FG / DF
12090   GOTO 12050
12100 CC = (1 - COS(GG)) / AI(Q)
12110 SS = SIN(GG) / SQR(ABS(AI(Q)))
12120 F = 1 - CC / R0(Q)
12130 G = (R0(Q) * SS + D0(Q) * CC) / SQR(1 + M(Q))
12140 R = R0(Q) + C0(Q) * CC + D0(Q) * SS
12150 FP = -SQR(1 + M(Q)) * SS / (R * R0(Q))
12160 GP = 1 - CC / R
12170 RETURN
13000 STOP
13010 REM # ENK5/SUB # RK5 NUMERICAL INTEGRATION
13020 REM # FOR THE MOTHOD OF ENCKE
13030 REM
13040 H = K# * DT
13050 REM
13060 REM ********** STEP 1 **************
13070 REM
13080 FOR K = 1 TO 3
13090   DV(0, K) = 0
13100   RP(0, K) = RQ(0, K)
```

```
13110  VP(0, K) = VQ(0, K)
13120 NEXT K
13130 REM
13140 FOR Q = 0 TO NQ
13150  FOR K = 1 TO 3
13160        RK(Q, K) = RQ(Q, K)
13170        VK(Q, K) = VQ(Q, K)
13180        PK(Q, K) = RK(Q, K) - RP(0, K)
13190  NEXT K
13200 NEXT Q
13210 REM
13220 GOSUB 15010: REM SUBSUB
13230 REM
13240 FOR K = 1 TO 3
13250  L1(K) = LX(K)
13260  S1(K) = SX(K)
13270  FOR Q = 0 TO NQ
13280        U1(Q, K) = UX(Q, K)
13290        W1(Q, K) = WX(Q, K)
13300  NEXT Q
13310  F1(K) = FX(K)
13320  G1(K) = GX(K)
13330 NEXT K
13340  REM
13350  REM ********** STEP 2 **************
13360  REM
13370  FOR K = 1 TO 3
13380        DV(0, K) = G1(K) / 4
13390        RP(0, K) = RQ(0, K) + L1(K) / 4
13400        VP(0, K) = VQ(0, K) + S1(K) / 4
13410  NEXT K
13420  REM
```

```
13430 FOR Q = 0 TO NQ
13440     FOR K = 1 TO 3
13450         RK(Q, K) = RQ(Q, K) + U1(Q, K) / 4
13460         VK(Q, K) = VQ(Q, K) + W1(Q, K) / 4
13470         PK(Q, K) = RK(Q, K) - RP(0, K)
13480     NEXT K
13490 NEXT Q
13500 REM
13510 GOSUB 15010: REM SUBSUB
13520 REM
13530 FOR K = 1 TO 3
13540     L2(K) = LX(K)
13550     S2(K) = SX(K)
13560     FOR Q = 0 TO NQ
13570         U2(Q, K) = UX(Q, K)
13580         W2(Q, K) = WX(Q, K)
13590     NEXT Q
13600     F2(K) = FX(K)
13610     G2(K) = GX(K)
13620 NEXT K
13630 REM
13640 REM ********* STEP 3 *********
13650 REM
13660 FOR K = 0 TO 3
13670     DV(0, K) = (G1(K) + G2(K)) / 8
13680     RP(0, K) = RQ(0, K) + (L1(K) + L2(K)) / 8
13690     VP(0, K) = VQ(0, K) + (S1(K) + S2(K)) / 8
13700 NEXT K
13710 REM
13720 FOR Q = 0 TO NQ
13730     FOR K = 1 TO 3
13740         RK(Q, K) = RQ(Q, K) + (U1(Q, K) + U2(Q, K)) / 8
```

```
13750            VK(Q, K) = VQ(Q, K) + (W1(Q, K) + W2(Q, K)) / 8
13760            PK(Q, K) = RK(Q, K) - RP(0, K)
13770        NEXT K
13780    NEXT Q
13790    REM
13800    GOSUB 15010: REM SUBSUB
13810    REM
13820    FOR K = 1 TO 3
13830        L3(K) = LX(K)
13840        S3(K) = SX(K)
13850        FOR Q = 0 TO NQ
13860            U3(Q, K) = UX(Q, K)
13870            W3(Q, K) = WX(Q, K)
13880        NEXT Q
13890        F3(K) = FX(K)
13900        G3(K) = GX(K)
13910    NEXT K
13920    REM
13930    REM ************** STEP 4 **************
13940    REM
13950    FOR K = 1 TO 3
13960        DV(0, K) = -(G2(K) - 2 * G3(K)) / 2
13970        RP(0, K) = RQ(0, K) - (L2(K) - 2 * L3(K)) / 2
13980        VP(0, K) = VQ(0, K) - (S2(K) - 2 * S3(K)) / 2
13990    NEXT K
14000    REM
14010    FOR Q = 0 TO NQ
14020        FOR K = 1 TO 3
14030            RK(Q, K) = RQ(Q, K) - (U2(Q, K) - 2 * U3(Q, K)) / 2
14040            VK(Q, K) = VQ(Q, K) - (W2(Q, K) - 2 * W3(Q, K)) / 2
14050            PK(Q, K) = RK(Q, K) - RP(0, K)
14060        NEXT K
```

```
14070  NEXT Q
14080  REM
14090  GOSUB 15010: REM SUBSUB
14100  REM
14110  FOR K = 1 TO 3
14120      L4(K) = LX(K)
14130      S4(K) = SX(K)
14140      FOR Q = 0 TO NQ
14150          U4(Q, K) = UX(Q, K)
14160          W4(Q, K) = WX(Q, K)
14170      NEXT Q
14180      F4(K) = FX(K)
14190      G4(K) = GX(K)
14200  NEXT K
14210  REM
14220  REM ************ STEP 5 **************
14230  REM
14240  FOR K = 1 TO 3
14250      DV(0, K) = (3 * G1(K) + 9 * G4(K)) / 16
14260      RP(0, K) = RQ(0, K) + (3 * L1(K) + 9 * L4(K)) / 16
14270      VP(0, K) = VQ(0, K) + (3 * S1(K) + 9 * S4(K)) / 16
14280  NEXT K
14290  REM
14300  FOR Q = 0 TO NQ
14310      FOR K = 1 TO 3
14320          RK(Q, K) = RQ(Q, K) + (3 * U1(Q, K) + 9 * U4(Q, K)) / 16
14330          VK(Q, K) = VQ(Q, K) + (3 * W1(Q, K) + 9 * W4(Q, K)) / 16
14340          PK(Q, K) = RK(Q, K) - RP(0, K)
14350      NEXT K
14360  NEXT Q
14370  REM
14380  GOSUB 15010: REM SUBSUB
```

```
14390  REM
14400  FOR K = 1 TO 3
14410      L5(K) = LX(K)
14420      S5(K) = SX(K)
14430      FOR Q = 0 TO NQ
14440          U5(Q, K) = UX(Q, K)
14450          W5(Q, K) = WX(Q, K)
14460      NEXT Q
14470      F5(K) = FX(K)
14480      G5(K) = GX(K)
14490  NEXT K
14500  REM
14510  REM ******* STEP 6 **********
14520  REM
14530  FOR K = 1 TO 3
14540      DV(O, K) = -(3 * G1(K) - 2 * G2(K) - 12 * G3(K) + 12 * G4(K) - 8 * G5(K)) / 7
14550      RP(0, K) = RQ(0, K) - (3 * L1(K) - 2 * L2(K) - 12 * L3(K) + 12 * L4(K) - 8 *
L5(K)) / 7
14560      VP(0, K) = VQ(0, K) - (3 * S1(K) - 2 * S2(K) - 12 * S3(K) + 12 * S4(K) - 8 *
S5(K))
/ 7
14570  NEXT K
14580  REM
14590  FOR Q = 0 TO NQ
14600      FOR K = 1 TO 3
14610          RK(Q, K) = RQ(Q, K) - (3 * U1(Q, K) - 2 * U2(Q, K) - 12 * U3(Q, K) + 12 *
U4(Q, K) - 8 * U5(Q, K)) / 7
14620          VK(Q, K) = VQ(Q, K) - (3 * W1(Q, K) - 2 * W2(Q, K) - 12 * W3(Q, K) + 12 *
W4(Q, K) - 8 * W5(Q, K)) / 7
14630          PK(Q, K) = RK(Q, K) - RP(0, K)
14640      NEXT K
14650  NEXT Q
```

```
14660  REM
14670  GOSUB 15010: REM SUBSUB
14680  REM
14690  FOR K = 1 TO 3
14700        F6(K) = FX(K)
14710        G6(K) = GX(K)
14720  NEXT K
14730  REM
14740  REM ************ RESULT ************
14750  REM
14760  FOR K = 1 TO 3
14770        DR(0, K) = (7 * F1(K) + 32 * F3(K) + 12 * F4(K) + 32 * F5(K) + 7 * F6(K)) / 90
14780        DV(0, K) = (7 * G1(K) + 32 * G3(K) + 12 * G4(K) + 32 * G5(K) + 7 * G6(K)) / 90
14790  NEXT K
14800 RETURN
15000 STOP
15010 REM # ENK5 SUBSUBROUTINE #
15020 REM
15030 FOR Q = 1 TO NQ
15040  RK(Q, 0) = FNMG(RK(Q, 1), RK(Q, 2), RK(Q, 3))
15050  PK(Q, 0) = FNMG(PK(Q, 1), PK(Q, 2), PK(Q, 3))
15060  FOR K = 1 TO 3
15070        AK(Q, K) = FNAK(M(Q), PK(Q, K), PK(Q, 0), RK(Q, K), RK(Q, 0))
15080  NEXT K
15090  NEXT Q
15100  REM
15110  FOR K = 1 TO 3
15120        AK(0, K) = 0
15130        FOR Q = 1 TO NQ
15140              AK(0, K) = AK(0, K) + AK(Q, K)
15150        NEXT Q
15160  NEXT K
```

```
15170  REM
15180  RK(0, 0) = FNMG(RK(0, 1), RK(0, 2), RK(0, 3))
15190  RP(0, 0) = FNMG(RP(0, 1), RP(0, 2), RP(0, 3))
15200  REM
15210  FOR K = 1 TO 3
15220      LX(K) = H * FNV(VP(0, K))
15230      SX(K) = H * FNA(M(0), RP(0, K), RP(0, 0), AK(0, K))
15240      FOR Q = 0 TO NQ
15250          UX(Q, K) = H * FNV(VK(Q, K))
15260          WX(Q, K) = H * FNA(M(Q), RK(Q, K), RK(Q, 0), 0)
15270      NEXT Q
15280      FX(K) = H * FNF(DV(0, K))
15290      GX(K) = H * FNG(M(0), RK(0, K), RK(0, 0), RP(0, K), RP(0, 0), AK(0, K))
15300  NEXT K
15310  RETURN
16000  STOP
16010  REM # ARC/SUB #COMPUTES X FROM SIN(X) AND COS(X)
16020  IF ABS(SX) <= .707107 THEN X = FNASN(ABS(SX))
16030  IF ABS(CX) <= .707107 THEN X = FNACN(ABS(CX))
16040  IF CX >= 0 AND SX >= 0 THEN X = X
16050  IF CX < 0 AND SX >= 0 THEN X = 180 * Q1# - X
16060  IF CX < 0 AND SX < 0 THEN X = 180 * Q1# + X
16070  IF CX >= 0 AND SX < 0 THEN X = 360 * Q1# - X
16080  RETURN
20000 DATA 0.005775519
20010 REM
20020 REM # celestial body #
20030 REM
30000 DATA "THULE","j2000.0",0.017202099
30010 DATA 0.0#
30020 DATA 12200.5
30030 DATA 2.0592292#,-3.4364384#,-1.6230714#
```

```
30040 DATA +0.4219005#,+0.2127084#,+0.0772689#
30050 REM
39000 REM # sun #
39010 REM
39020 DATA 21820.5#,-0.8672422#,+0.4774827#,+0.2070265#
39030 REM
40000 REM # perturbers #
40010 REM
40020 'data 0,"( two body )",0
40030 DATA 2,"(js)",12200.5#
40040 REM
45000 DATA"jupiter"
45010 DATA 0.000954791#
45020 DATA 0.4502857#,4.6806179#,1.9953986#
45030 DATA -0.007613625#,+0.000871374#,+0.0005587947#
46000 DATA"saturn"
46010 DATA 0.000285878#
46020 DATA -8.1446653#,3.5712563#,1.8256609#
46030 DATA -0.0024447485#,-0.004948238824#,-0.001938251665#
```

ภาคผนวก ง

โปรแกรมการคำนวณหลักมูลทางโคจรจากตำแหน่งและความเร็ว

```
1000  CLS
1010  PRINT "#classel#classical elements"
1020  PRINT "#from pocition and velocity"
1030  REM
1040  DEFDBL a-z
1050  DEFINT k
1060  REM
1070  DEF fnvs (x, y, z) = x * x + y * y + z * z
1080  DEF fnmg (x, y, z) = SQR(x * x + y * y + z * z)
1090  DEF fndp (x1, y1, z1, x2, y2, z2) = x1 * x2 + y1 * y2 + z1 * z2
1100  DEF fnasn (x) = ATN(x / SQR(-x * x + 1))
1110  DEF fnacn (x) = -ATN(x / SQR(-x * x + 1) + 1.5707963263#)
1120  DEF fngsh (x) = LOG(x + SQR(x * x + 1))
1130  REM
1140  REM
1150  q1 = .0174532926#
1160  pi = 3.1415926536#
1170  g$ = "#######.#######"
1180  s$ = "#########.#####"
1190  REM
1200  READ ec
1210  READ n$, e$, k#
1220  READ m
1230  READ t0
1240  READ r0(1), r0(2), r0(3)
1250  READ v0(1), v0(2), v0(3)
1260  REM
1270  LINE INPUT " "; a$
```

```
1280  CLS
1290  PRINT "classical element"
1300  PRINT "from"
1310  PRINT "position and velocity"
1320  PRINT n$
1330  PRINT e$
1340  PRINT "k"; TAB(7); : PRINT USING g$; k#
1350  PRINT "ec"; TAB(7); : PRINT USING g$; ec
1360  PRINT
1370  PRINT "t(0)"; TAB(7); : PRINT USING s$; t0
1380  PRINT
1390  PRINT "position and velocity"
1400  PRINT
1410  PRINT "r(k)"; TAB(7);
1420  PRINT USING g$; r0(1); r0(2); r0(3)
1430  PRINT "v(k)"; TAB(7);
1440  PRINT USING g$; v0(1); v0(2); v0(3)
1450  REM
1460  REM
1470  REM
1480  REM***reduction to ecliptic if required***
1490  REM
1500  ec = ec * q1
1510  REM
1520  r(1) = r0(1)
1530  r(2) = r0(2) * COS(ec) + r0(3) * SIN(ec)
1540  r(3) = r0(3) * COS(ec) - r0(2) * SIN(ec)
1550  REM
1560  v(1) = v0(1)
1570  v(2) = v0(2) * COS(ec) + v0(3) * SIN(ec)
1580  v(3) = v0(3) * COS(ec) - v0(2) * SIN(ec)
1590  REM
```

```
1600  REM***vector e(k),h(k),n(k)***
1610  REM
1620  r = fnmg(r(1), r(2), r(3))
1630  v2 = fnvs(v(1), v(2), v(3))
1640  rv = fndp(r(1), r(2), r(3), v(1), v(2), v(3))
1650  REM
1660  FOR k = 1 TO 3
1670  e(k) = (v2 / m - 1 / r) * r(k) - (rv / m) * v(k)
1680  NEXT k
1690  REM
1700  h(1) = r(2) * v(3) - r(3) * v(2)
1710  h(2) = r(3) * v(1) - r(1) * v(3)
1720  h(3) = r(1) * v(2) - r(2) * v(1)
1730  REM
1740  nn(1) = -h(2)
1750  nn(2) = h(1)
1760  nn(3) = 0
1770  REM
1780  REM***element a,e and q***
1790  REM
1800  ai = 2 / r - v2 / m
1810  e = fnmg(e(1), e(2), e(3))
1820  sp = fnvs(h(1), h(2), h(3)) / m
1830  q = sp / (1 + e)
1840  REM
1850  REM***element i,oo,w***
1860  REM
1870  h = fnmg(h(1), h(2), h(3))
1880  i = fnacn(h(3) / h) / q1
1890  REM
1900  nn = fnmg(nn(1), nn(2), nn(3))
1910  oo = fnacn(nn(1) / nn) / q1
```

```
1920  IF nn(2) < 0 THEN oo = 360 - oo
1930  REM
1940  ne = fndp(nn(1), nn(2), nn(3), e(1), e(2), e(3))
1950  w = fnacn(ne / (nn * e)) / q1
1960  IF e(3) < 0 THEN w = 360 - w
1970  REM
2500  REM
2510  LINE INPUT " "; a$
2520  CLS
2530  PRINT "classical elements"
2540  PRINT
2550  PRINT aq$; TAB(7); : PRINT USING g$; aq
2560  PRINT "e"; TAB(7); : PRINT USING g$; e
2570  'PRINT mt$; TAB(7); : PRINT USING s$; mt
2580  PRINT "i"; TAB(7); : PRINT USING s$; i
2590  PRINT "oo"; TAB(7); : PRINT USING s$; oo
2600  PRINT "w"; TAB(7); : PRINT USING s$; w
2610  PRINT
2620  REM
2680  END
20000  DATA 23.439291#
20010  DATA"thule","j2000.0",0.017202099#
20020  DATA  1
20030  DATA 2452200.5#
20040  DATA 4.2389113#,-0.7168452#,-0.4994156#
20050  DATA 0.0932830#,0.4296750#,0.1881310#
20060  REM
```

ประวัติผู้เขียนวิทยานิพนธ์

นางสาวสุนทรี พงศ์รักธรรม ศึกษาชั้นประถมศึกษาและมัธยมศึกษาจากโรงเรียนเทศบาลวัดชัยชุมพลและโรงเรียนสตรีทุ่งสง จังหวัดนครศรีธรรมราช ตามลำดับ สำเร็จการศึกษาปริญญาตรีสาขาวิชาฟิสิกส์ คณะวิทยาศาสตร์ มหาวิทยาลัยสงขลานครินทร์ในปี 2543 ต่อมาเข้ารับการศึกษาต่อในระดับปริญญาโทในสาขาวิชาฟิสิกส์ ณ จุฬาลงกรณ์มหาวิทยาลัยในปี 2544